NAG Library Routine Document

F08HNF (ZHBEV)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08HNF (ZHBEV) computes all the eigenvalues and, optionally, all the eigenvectors of a complex n by n Hermitian band matrix A of bandwidth $(2k_d + 1)$.

2 Specification

```
SUBROUTINE F08HNF (JOBZ, UPLO, N, KD, AB, LDAB, W, Z, LDZ, WORK, RWORK,    &
                   INFO)

INTEGER                 N, KD, LDAB, LDZ, INFO
REAL (KIND=nag_wp)      W(N), RWORK(3*N-2)
COMPLEX (KIND=nag_wp) AB(LDAB,*), Z(LDZ,*), WORK(N)
CHARACTER(1)            JOBZ, UPLO
```

The routine may be called by its LAPACK name ***zhbev***.

3 Description

The Hermitian band matrix A is first reduced to real tridiagonal form, using unitary similarity transformations, and then the QR algorithm is applied to the tridiagonal matrix to compute the eigenvalues and (optionally) the eigenvectors.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Arguments

1: JOBZ – CHARACTER(1) *Input*

On entry: indicates whether eigenvectors are computed.

JOBZ = 'N'
: Only eigenvalues are computed.

JOBZ = 'V'
: Eigenvalues and eigenvectors are computed.

Constraint: JOBZ = 'N' or 'V'.

2: UPLO – CHARACTER(1) *Input*

On entry: if UPLO = 'U', the upper triangular part of A is stored.

If UPLO = 'L', the lower triangular part of A is stored.

Constraint: UPLO = 'U' or 'L'.

3: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

4: KD – INTEGER *Input*

On entry: if UPLO = 'U', the number of superdiagonals, k_d, of the matrix A.

If UPLO = 'L', the number of subdiagonals, k_d, of the matrix A.

Constraint: $\mathrm{KD} \geq 0$.

5: AB(LDAB, $*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array AB must be at least $\max(1, \mathrm{N})$.

On entry: the upper or lower triangle of the n by n Hermitian band matrix A.

The matrix is stored in rows 1 to $k_d + 1$, more precisely,

if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element A_{ij} in $\mathrm{AB}(k_d + 1 + i - j, j)$ for $\max(1, j - k_d) \leq i \leq j$;

if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element A_{ij} in $\mathrm{AB}(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_d)$.

On exit: AB is overwritten by values generated during the reduction to tridiagonal form.

The first superdiagonal or subdiagonal and the diagonal of the tridiagonal matrix T are returned in AB using the same storage format as described above.

6: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F08HNF (ZHBEV) is called.

Constraint: $\mathrm{LDAB} \geq \mathrm{KD} + 1$.

7: W(N) – REAL (KIND=nag_wp) array *Output*

On exit: the eigenvalues in ascending order.

8: Z(LDZ, $*$) – COMPLEX (KIND=nag_wp) array *Output*

Note: the second dimension of the array Z must be at least $\max(1, \mathrm{N})$ if JOBZ = 'V', and at least 1 otherwise.

On exit: if JOBZ = 'V', Z contains the orthonormal eigenvectors of the matrix A, with the ith column of Z holding the eigenvector associated with $\mathrm{W}(i)$.

If JOBZ = 'N', Z is not referenced.

9: LDZ – INTEGER *Input*

On entry: the first dimension of the array Z as declared in the (sub)program from which F08HNF (ZHBEV) is called.

Constraints:

if JOBZ = 'V', $\mathrm{LDZ} \geq \max(1, \mathrm{N})$;
otherwise $\mathrm{LDZ} \geq 1$.

10: WORK(N) – COMPLEX (KIND=nag_wp) array *Workspace*

11: RWORK($3 \times \mathrm{N} - 2$) – REAL (KIND=nag_wp) array *Workspace*

12: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, the algorithm failed to converge; i off-diagonal elements of an intermediate tridiagonal form did not converge to zero.

7 Accuracy

The computed eigenvalues and eigenvectors are exact for a nearby matrix $(A + E)$, where

$$\|E\|_2 = O(\epsilon)\|A\|_2,$$

and ϵ is the ***machine precision***. See Section 4.7 of Anderson *et al.* (1999) for further details.

8 Parallelism and Performance

F08HNF (ZHBEV) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08HNF (ZHBEV) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The total number of floating-point operations is proportional to n^3 if JOBZ = 'V' and is proportional to $k_d n^2$ otherwise.

The real analogue of this routine is F08HAF (DSBEV).

10 Example

This example finds all the eigenvalues and eigenvectors of the Hermitian band matrix

$$A = \begin{pmatrix} 1 & 2-i & 3-i & 0 & 0 \\ 2+i & 2 & 3-2i & 4-2i & 0 \\ 3+i & 3+2i & 3 & 4-3i & 5-3i \\ 0 & 4+2i & 4+3i & 4 & 5-4i \\ 0 & 0 & 5+3i & 5+4i & 5 \end{pmatrix},$$

together with approximate error bounds for the computed eigenvalues and eigenvectors.

10.1 Program Text

```
    Program f08hnfe

!     F08HNF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: ddisna, dznrm2, nag_wp, x02ajf, x04daf, zhbev
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter              :: nin = 5, nout = 6
      Character (1), Parameter        :: uplo = 'U'
!     .. Local Scalars ..
      Complex (Kind=nag_wp)           :: scal
      Real (Kind=nag_wp)              :: eerrbd, eps
      Integer                         :: i, ifail, info, j, k, kd, ldab, ldz, &
                                         n
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ab(:,:), work(:), z(:,:)
      Real (Kind=nag_wp), Allocatable  :: rcondz(:), rwork(:), w(:), zerrbd(:)
!     .. Intrinsic Procedures ..
      Intrinsic                       :: abs, conjg, max, maxloc, min
!     .. Executable Statements ..
      Write (nout,*) 'F08HNF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kd
      ldab = kd + 1
      ldz = n
      Allocate (ab(ldab,n),work(n),z(ldz,n),rcondz(n),rwork(3*n-2),w(n),      &
        zerrbd(n))

!     Read the upper or lower triangular part of the symmetric band
!     matrix A from data file

      If (uplo=='U') Then
        Read (nin,*)((ab(kd+1+i-j,j),j=i,min(n,i+kd)),i=1,n)
      Else If (uplo=='L') Then
        Read (nin,*)((ab(1+i-j,j),j=max(1,i-kd),i),i=1,n)
      End If

!     Solve the band Hermitian eigenvalue problem
!     The NAG name equivalent of zhbev is f08hnf
      Call zhbev('Vectors',uplo,n,kd,ab,ldab,w,z,ldz,work,rwork,info)

      If (info==0) Then

!       Print solution

        Write (nout,*) 'Eigenvalues'
        Write (nout,99999) w(1:n)
        Flush (nout)

!       Normalize the eigenvectors, largest element real
        Do i = 1, n
          rwork(1:n) = abs(z(1:n,i))
          k = maxloc(rwork(1:n),1)
          scal = conjg(z(k,i))/abs(z(k,i))/dznrm2(n,z(1,i),1)
          z(1:n,i) = z(1:n,i)*scal
        End Do

!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        Call x04daf('General',' ',n,n,z,ldz,'Eigenvectors',ifail)

!       Get the machine precision, EPS and compute the approximate
```

```
!       error bound for the computed eigenvalues.  Note that for
!       the 2-norm, max( abs(W(i)) ) = norm(A), and since the
!       eigenvalues are returned in ascending order
!       max( abs(W(i)) ) = max( abs(W(1)), abs(W(n)))

        eps = x02ajf()
        eerrbd = eps*max(abs(w(1)),abs(w(n)))

!       Call DDISNA (F08FLF) to estimate reciprocal condition
!       numbers for the eigenvectors
        Call ddisna('Eigenvectors',n,n,w,rcondz,info)

!       Compute the error estimates for the eigenvectors

        Do i = 1, n
          zerrbd(i) = eerrbd/rcondz(i)
        End Do

!       Print the approximate error bounds for the eigenvalues
!       and vectors

        Write (nout,*)
        Write (nout,*) 'Error estimate for the eigenvalues'
        Write (nout,99998) eerrbd
        Write (nout,*)
        Write (nout,*) 'Error estimates for the eigenvectors'
        Write (nout,99998) zerrbd(1:n)
      Else
        Write (nout,99997) 'Failure in ZHBEV. INFO =', info
      End If

99999 Format (3X,(8F8.4))
99998 Format (4X,1P,6E11.1)
99997 Format (1X,A,I4)
    End Program f08hnfe
```

10.2 Program Data

```
F08HNF Example Program Data

   5          2                                          :Values of N and KD

  (1.0, 0.0) (2.0,-1.0) (3.0,-1.0)
             (2.0, 0.0) (3.0,-2.0) (4.0,-2.0)
                        (3.0, 0.0) (4.0,-3.0) (5.0,-3.0)
                                   (4.0, 0.0) (5.0,-4.0)
                                              (5.0, 0.0) :End of matrix A
```

10.3 Program Results

```
 F08HNF Example Program Results

 Eigenvalues
    -6.4185 -1.4094  1.4421  4.4856 16.9002
 Eigenvectors
          1       2       3       4       5
 1  -0.2534  0.6367 -0.2560  0.0171  0.1051
    -0.0538  0.0000  0.3721  0.5500 -0.0983

 2  -0.0662 -0.2578  0.5344 -0.2608  0.2516
     0.4301  0.2413  0.0000  0.4869 -0.1789

 3   0.5274 -0.3039 -0.4245 -0.0399  0.4994
     0.0000 -0.3481  0.0915  0.2142 -0.1513

 4   0.1061  0.3450  0.4964 -0.0253  0.5611
    -0.4981 -0.0832 -0.1546 -0.1700  0.0000

 5  -0.4519 -0.2469 -0.1979  0.5614  0.4837
     0.0424  0.2634 -0.1114 -0.0000  0.2509
```

```
 Error estimate for the eigenvalues
        1.9E-15

 Error estimates for the eigenvectors
        3.7E-16    6.6E-16    6.6E-16    6.2E-16    1.5E-16
```